

Stars

In this game, the computer selects a random number from 1 to 100 (or any value you set in Statement 150). You try to guess the number and the computer gives you clues to tell you how close you're getting. One star (*) means you're far away from the number; seven stars (*****) means you're really close. You get 7 guesses.

On the surface this game is very similar to GUESS; however, the guessing strategy is quite different. See if you can come up with one or more approaches to finding the mystery number.

Bob Albrecht of People's Computer Company created this game.

```

                STARS
CREATIVE COMPUTING MORRISTOWN, NEW JERSEY

```

```

DO YOU WANT INSTRUCTIONS? YES
I AM THINKING OF A WHOLE NUMBER FROM 1 TO 100
TRY TO GUESS MY NUMBER. AFTER YOU GUESS, I
WILL TYPE ONE OR MORE STARS (*). THE MORE
STARS I TYPE, THE CLOSER YOU ARE TO MY NUMBER.
ONE STAR (*) MEANS FAR AWAY, SEVEN STARS (*****)
MEANS REALLY CLOSE! YOU GET 7 GUESSES.

```

OK, I AM THINKING OF A NUMBER, START GUESSING.

```

YOUR GUESS? 50
***

```

```

YOUR GUESS? 25
*****!!!!
YOU GOT IT IN 2 GUESSES!!! LET'S PLAY AGAIN...

```

OK, I AM THINKING OF A NUMBER, START GUESSING.

```

YOUR GUESS? 50
***

```

```

YOUR GUESS? 75
**

```

```

YOUR GUESS? 25
*****

```

```

YOUR GUESS? 15
***

```

```

YOUR GUESS? 35
*****

```

```

YOUR GUESS? 38
*****

```

```

YOUR GUESS? 33
*****

```

SORRY, THAT'S 7 GUESSES, NUMBER WAS 32

OK, I AM THINKING OF A NUMBER, START GUESSING.

```

YOUR GUESS? 50
***

```

```

YOUR GUESS? 75
**

```

```

YOUR GUESS? 25
*****

```

```

YOUR GUESS? 35
*****

```

```

YOUR GUESS? 30
*****

```

```

YOUR GUESS? 31
*****!!!!
YOU GOT IT IN 6 GUESSES!!! LET'S PLAY AGAIN...

```

```

10 PRINT TAB(34);"STARS"
20 PRINT TAB(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
30 PRINT:PRINT:PRINT
100 REM *** STARS - PEOPLE'S COMPUTER CENTER, MENLO PARK, CA
140 REM *** A IS LIMIT ON NUMBER, M IS NUMBER OF GUESSES
150 A=100:M=7
170 INPUT "DO YOU WANT INSTRUCTIONS?";A$
190 IF LEFT$(A$,1)="N" THEN 280
200 REM *** INSTRUCTIONS ON HOW TO PLAY
210 PRINT "I AM THINKING OF A WHOLE NUMBER FROM 1 TO";A
220 PRINT "TRY TO GUESS MY NUMBER. AFTER YOU GUESS, I"
230 PRINT "WILL TYPE ONE OR MORE STARS (*). THE MORE"
240 PRINT "STARS I TYPE, THE CLOSER YOU ARE TO MY NUMBER."
250 PRINT "ONE STAR (*) MEANS FAR AWAY, SEVEN STARS (*****)"
260 PRINT "MEANS REALLY CLOSE! YOU GET";M;"GUESSES."
270 REM *** COMPUTER THINKS OF A NUMBER
280 PRINT
290 PRINT
300 X=INT(A*RND(1)+1)
310 PRINT "OK, I AM THINKING OF A NUMBER, START GUESSING."
320 REM *** GUESSING BEGINS, HUMAN GETS M GUESSES
330 FOR K=1 TO M
340 PRINT
350 PRINT "YOUR GUESS";
360 INPUT G
370 IF G=X THEN 600
380 D=ABS(G-X)
390 IF D>=64 THEN 510
400 IF D>=32 THEN 500
410 IF D>=16 THEN 490
420 IF D>=8 THEN 480
430 IF D>=4 THEN 470
440 IF D>=2 THEN 460
450 PRINT "*";
460 PRINT "*";
470 PRINT "*";
480 PRINT "*";
490 PRINT "*";
500 PRINT "*";
510 PRINT "*";
520 PRINT
530 NEXT K
540 REM *** DID NOT GUESS IN M GUESSES
550 PRINT
560 PRINT "SORRY, THAT'S";M;"GUESSES, NUMBER WAS";X
580 GOTO 280
590 REM *** WE HAVE A WINNER
600 FOR N=1 TO 50
610 PRINT "*";
620 NEXT N
630 PRINT "!!!!"
640 PRINT "YOU GOT IT IN";K;"GUESSES!!! LET'S PLAY AGAIN..."
650 GOTO 280
660 END

```